F03AFF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F03AFF computes an LU factorization of a real matrix, with partial pivoting, and evaluates the determinant.

2 Specification

```
SUBROUTINE F03AFF(N, EPS, A, IA, D1, ID, P, IFAIL)
INTEGER           N, IA, ID, IFAIL
real              EPS, A(IA,*), D1, P(*)
```

3 Description

This routine computes an LU factorization of a real matrix A with partial pivoting: $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. The determinant of A is the product of the diagonal elements of L with the correct sign determined by the row interchanges.

4 References

[1] Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

5 Parameters

1: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

2: EPS — ***real*** *Input*

On entry: EPS must be set to the value of

machine precision.

3: A(IA,∗) — ***real*** array *Input/Output*

Note: the second dimension of the array A must be at least max(1,N).

On entry: the n by n matrix A.

On exit: A is overwritten by the lower triangular matrix L and the off-diagonal elements of the upper triangular matrix U. The unit diagonal elements of U are not stored.

4: IA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03AFF is called.

Constraint: IA $\geq$ max(1,N).

5: D1 — ***real*** *Output*

6: ID — INTEGER *Output*

On exit: the determinant of A is given by D1 $\times 2.0^{\mathrm{ID}}$. It is given in this form to avoid overflow or underflow.

7: P(∗) — ***real*** array *Output*

Note: the dimension of the array P must be at least max(1,N).

On exit: P(i) gives the row index of the ith pivot.

8: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = 1

The matrix A is singular, possibly due to rounding errors. The factorization could not be completed. D1 and ID are set to zero.

IFAIL = 2

On entry, N < 0,
or IA < max(1,N).

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis, see Wilkinson and Reinsch [1] page 107.

8 Further Comments

The time taken by the routine is approximately proportional to n^3.

9 Example

To compute the LU factorization with partial pivoting, and calculate the determinant, of the real matrix:

$$\begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F03AFF Example Program Text
*     Mark 15 Revised.  NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NMAX, IA
      real             TWO
      PARAMETER        (NMAX=8,IA=NMAX,TWO=2.0e0)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
```

```
*     .. Local Scalars ..
      real             D1
      INTEGER          I, ID, IFAIL, J, N
*     .. Local Arrays ..
      real             A(IA,NMAX), P(NMAX)
*     .. External Functions ..
      real             X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         F03AFF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F03AFF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         IFAIL = 0
*
         CALL F03AFF(N,X02AJF(),A,IA,D1,ID,P,IFAIL)
*
         WRITE (NOUT,*) 'Array A after factorization'
         DO 20 I = 1, N
            WRITE (NOUT,99998) (A(I,J),J=1,N)
   20    CONTINUE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Array P'
         WRITE (NOUT,99998) (P(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,99997) 'D1 = ', D1, ' ID = ', ID
         D1 = D1*TWO**ID
         WRITE (NOUT,*)
         WRITE (NOUT,99997) 'Value of determinant = ', D1
      ELSE
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,A,F9.4,A,I2)
      END
```

9.2 Program Data

```
F03AFF Example Program Data
  3
   33   16   72
  -24  -10  -57
   -8   -4  -17
```

9.3 Program Results

```
F03AFF Example Program Results

Array A after factorization
  -8.0000   0.5000   2.1250
 -24.0000   2.0000  -3.0000
  33.0000  -0.5000   0.3750

Array P
   3.0000   2.0000   3.0000

D1 =    0.3750 ID =  4

Value of determinant =    6.0000
```